## Уравнение вертикального движения ракеты

In[266]:=
```
eq = (M0 - μ t) x''[t] == -(M0 - μ t) g + μ u;
```

$M_0$ - масса Лунолёта в момент $t = t_0$ (пусть масса без топлива будет 1000.0 (сухая масса), а начальный запас топлива 100 кг)

$\mu = \frac{\Delta m}{\Delta t}$ - секундный расход топлива (кг/с)

Δm - расход топлива (задается пользователем)

Δt - продолжительность расхода топлива (задается пользователем)

g - ускорение свободного падения на Луне (предположим, что это ускорение не зависит от высоты и равно 1.622 м/с$^2$

u - скорость истечения продуктов сгорания (пусть будет 3000 м/с - ЖРД). Положительное значение соответсвует  направлению силы тяги вверх (взлет).

In[278]:=
```
FSim[f_] := FullSimplify[f /. {t → Δt, μ → Δm/Δt, V0 → V0},
   {Δm < M0, Δm > 0, Δt > 0, g > 0, h0 > 0, t0 > 0, M0 > 0}];
```

## Интегрируем с начальными условиями x[0] = h$_0$, x'[0] = V$_0$

In[267]:=
```
solution = DSolve[{eq, x[0] == h0, x'[0] == V0}, x[t], t][[1]];
```

Новое значение высоты $h(t_k)$ в момент времени $t_k = t_0 + \Delta t$ (расход топлива не равен нулю):

In[279]:=
```
hk = FSim[x[t] /. solution]
```

Out[279]=
$$-\frac{\Delta t^2 g}{2} + h_0 - \Delta t\, u \log\left(1 - \frac{\Delta m}{M_0}\right) + \frac{\Delta t\, M_0\, u \log\left(1 - \frac{\Delta m}{M_0}\right)}{\Delta m} + \Delta t\, u + \Delta t\, V_0$$

log -- натуральный логарифм

Новое значение скорости $V(\Delta t)$ в
момент времени $t = \Delta t$ (расход топлива не равен нулю) :

In[280]:=
```
Vk = FSim[D[x[t] /. solution, t]]
```

Out[280]=
$$\Delta t\,(-g) - 2\, u \tanh^{-1}\left(\frac{\Delta m}{\Delta m - 2 M_0}\right) + V_0$$

Если расход топлива равен нулю, то движение происходит только под действием силы тяжести. Через Δt высота будет равна:

In[281]:=
```
Limit[FSim[x[t] /. solution], Δm → 0]
```

Out[281]=
$$-\frac{\Delta t^2 g}{2} + h_0 + \Delta t\, V_0$$

Если расход топлива равен нулю, то движение происходит только под действием силы тяжести. Через Δt скорость будет равна:

In[282]:=
```
Limit[FSim[D[x[t] /. solution, t]], Δm → 0]
```

Out[282]=
$$V_0 - \Delta t\, g$$

## Перегрузка Лунолета при работе двигателя (приблизительно)

$n_x \simeq \mu u / (M_0 * 9.81)$

In[276]:=
```
hk /. {Δt → 1.0, u → 3000.0, M0 → 1000.0, Δm → 1, V0 → 0.0, h0 → 1000.0, g → 1.622}
Vk /. {Δt → 1.0, u → 3000.0, M0 → 1000.0, Δm → 1, V0 → 0.0, h0 → 1000.0, g → 1.622}
```

Out[276]= 1000.69

Out[277]= 1.3795